

EE 2004
Week 7 Tutorial
Solution

1. Indirect addressing mode: Use of the file select registers (FSR)

Write a program to copy the contents in data memory block from address 0x020 to 0x07F to the data memory block starting from address 0x130 to 0x18F by using FSRs.

```
LIST P=18F4520    ;directive to define processor
#include <P18F4520.INC> ;processor specific
                    ;variable definitions

;-----
Block1StartingAddress equ 0x020
Block2StartingAddress equ 0x130

                    ORG      0x000000
                    goto Initialize

                    ; Part I: You put something in file reg 020 to 07F
                    ; to test the code you have written in Part II.
                    ; If you were asked this question in exam, you do
                    ; not need to include Part I.
Initialize:        lfsr 0, Block1StartingAddress ; Fill 0xFF into address
                                                ;ranges 020 to 07F
LoopInitialize:    movlw 0xFF
                    movwf POSTINC0
                    movlw 0x7F
                    cpfsgt FSR0L; if [FSR0L] > 7F, skip next line and
                    ;enter Part II.
                    bra LoopInitialize

                    ; Part II: Copy contents in 020:07F to 130:18F

                    lfsr 0, Block1StartingAddress
                    lfsr 1, Block2StartingAddress;
Copy:              movff POSTINC0, POSTINC1
                    movlw 0x7F
                    cpfsgt FSR0L
                    bra Copy
Over:              bra $
END
```

2. Table lookup

Define an array with starting address 0x000500 in the program memory to store your student ID number. Each digit of your student ID number is stored in a byte. Then, write a program to copy the content from the program memory to the data memory starting from 0x040.

```
LIST P=18F4520 ;directive to define processor
#include <P18F4520.INC>;processor specific
; variable definitions

;-----
COUNT equ 0x00
;-----
ORG 0x0000
;-----
;Start of main program

Main:    movlw upper ID; identify upper byte part of the starting
        ; address of the array labelled ID
        movwf TBLPTRU
        movlw high ID; identify high byte part of the starting
        ; address of the array labelled ID
        movwf TBLPTRH
        movlw low ID; identify low byte part of the starting
        ; address of the array labelled ID
        movwf TBLPTRL
        movlw 0x08
        movwf COUNT, A
        lfsr 0, 0x040
Loop:    tblrd*+
        movff TABLAT, POSTINC0
        decfsz COUNT, F, A
        bra Loop

        bra $

        org 0x000500
ID db 0x0, 0x1, 0x2, 0x3, 0x4, 0x5, 0x6, 0x7

END
```